

The worker/wrapper transformation

Jaak Randmets

April 20, 2010

Introduction

- The worker/wrapper transformation is a technique for changing the type of computation.
- Usually the aim is to improve performance by improving the choice of data structures used.
- Well known to compiler writers, but not so in the functional programming community.
- In this talk we provide systematic recipe for its use and explore it using wide range of examples.

The basic idea

Given some (recursive) function:

$$f = \mathbf{body}$$

the first step is to apply appropriate functions *wrap* and *unwrap* that allow the function *f* to be redefined by equation

$f = \mathit{wrap} (\mathit{unwrap} \mathbf{body})$. Next step is to split the function into two by naming the intermediate result:

$$\begin{aligned} f &= \mathit{wrap} \mathit{work} \\ \mathit{work} &= \mathit{unwrap} \mathbf{body} \end{aligned}$$

We then eliminate the mutual recursion:

$$\begin{aligned} f &= \mathit{wrap} \mathit{work} \\ \mathit{work} &= \mathit{unwrap} (\mathbf{body} [\mathit{wrap} \mathit{work} / f]) \end{aligned}$$

The worker/wrapper transformation

- We begin by defining fixed point operator in Haskell:

$$\begin{aligned} \text{fix} &:: (a \rightarrow a) \rightarrow a \\ \text{fix } f &= f (\text{fix } f) \end{aligned}$$

- In order to formalize the worker/wrapper transformation we will use property of fixed points known as the *rolling rule*:

$$\text{fix } (g \circ f) = g (\text{fix } (f \circ g))$$

Intuitively this is correct because both sides expand to the application $g (f (g (f \dots)))$.

The worker/wrapper transformation

- Supposed we have computation defined as a fixed point:

$comp :: A$
 $comp = fix\ body$
 $body :: A \rightarrow A$

We wish to change the underlying type A to some other type B .

- Worker/wrapper approach to this problem is to define conversion functions:

$unwrap :: A \rightarrow B$
 $wrap \quad :: B \rightarrow A$

such that $wrap \circ unwrap = id$.

The worker/wrapper transformation

comp
= { applying *comp* }
 fix body
= { *id* is identity for \circ }
 fix (id \circ body)
= { assuming *wrap* \circ *unwrap* = *id* }
 *fix (wrap \circ *unwrap* \circ *body*)*
= { use of rolling rule }
 *wrap (fix (unwrap \circ *body* \circ *wrap*))*
= { define *work* = *fix (unwrap \circ *body* \circ *wrap*)* }
 wrap work

The worker/wrapper transformation

Sometimes we might require a weaker property. Any of the following assumptions is valid:

Worker/wrapper assumptions

- $wrap \circ unwrap = id$
- $wrap \circ unwrap \circ body = body$
- $fix (wrap \circ unwrap \circ body) = fix body$

In general it's not the case that $unwrap \circ wrap = id$. However, following fusion property holds:

Worker/wrapper fusion

If $wrap \circ unwrap = id$, then $unwrap (wrap work) = work$.

Can easily be shown by taking:

$$work = unwrap (body (wrap work))$$

The worker/wrapper transformation

The worker/wrapper transformation

If $comp :: A$ is a recursive computation defined by $comp = fix\ body$ for some $body :: A \rightarrow A$, and $wrap :: B \rightarrow A$ and $unwrap :: A \rightarrow B$ are conversion functions satisfying any of the worker/wrapper assumptions, then:

$$comp = wrap\ work$$

where $work :: B$ is defined by:

$$work = fix\ (unwrap \circ body \circ wrap)$$

Difference lists

- Difference lists is an alternative way of representing lists:

type $DList\ a = [a] \rightarrow [a]$

- Naturally we need to convert to and from difference lists:

$fromList \quad :: [a] \rightarrow DList\ a$

$fromList\ xs = (xs \mathrel{++})$

$toList \quad \quad :: DList\ a \rightarrow [a]$

$toList\ f \quad = f\ []$

- We observe an identity $toList \circ fromList = id$:

$(toList \circ fromList)\ xs = xs \mathrel{++} [] = xs$

Difference lists

- Important property of difference lists is that *fromList* forms a morphism from lists to functions:

$$\begin{aligned} \text{fromList } (xs \mathrel{++} ys) &= \text{fromList } xs \circ \text{fromList } ys \\ \text{fromList } [] &= \text{id} \end{aligned}$$

- We can verify this by simple calculations:

$$\begin{aligned} \text{fromList } (xs \mathrel{++} ys) \text{ } zs &= (xs \mathrel{++} ys) \mathrel{++} zs \\ &= xs \mathrel{++} (ys \mathrel{++} zs) \\ &= \text{fromList } xs \text{ } (ys \mathrel{++} zs) \\ &= \text{fromList } xs \text{ } (\text{fromList } ys \text{ } zs) \\ &= (\text{fromList } xs \circ \text{fromList } ys) \text{ } zs \end{aligned}$$

$$\begin{aligned} \text{fromList } [] \text{ } zs &= [] \mathrel{++} zs \\ &= zs \end{aligned}$$

Reverse

- Let us consider the following definition of reverse:

$$\begin{aligned} rev &:: [a] \rightarrow [a] \\ rev [] &= [] \\ rev (x : xs) &= rev xs ++ [x] \end{aligned}$$

- As a first step we redefine *rev* as a fixed point

$$\begin{aligned} rev &:: [a] \rightarrow [a] \\ rev &= \text{fix } body \\ body &:: ([a] \rightarrow [a]) \rightarrow ([a] \rightarrow [a]) \\ body f [] &= [] \\ body f (x : xs) &= f xs ++ [x] \end{aligned}$$

Reverse

- Our aim is to change to computation type $[a] \rightarrow [a]$ to a type $[a] \rightarrow DList\ a$:

$unwrap :: ([a] \rightarrow [a]) \rightarrow ([a] \rightarrow DList\ a)$

$unwrap\ f = fromList \circ f$

$wrap :: ([a] \rightarrow DList\ a) \rightarrow ([a] \rightarrow [a])$

$wrap\ g = toList \circ g$

- We can verify the worker/wrapper assumption by:

$$\begin{aligned}(wrap \circ unwrap)\ f &= wrap\ (unwrap\ f) \\ &= toList \circ unwrap\ f \\ &= toList \circ fromList \circ f \\ &= id \circ f \\ &= f\end{aligned}$$

Reverse

Simplifying *rev*

rev :: $[a] \rightarrow [a]$

rev = *wrap work*

work :: $[a] \rightarrow DList\ a$

work = *unwrap* (*body* (*wrap work*))

- Inline *wrap* in body of *rev*.
- η -expand *work*.

Reverse

Simplifying *rev*

```
rev      :: [a] → [a]
rev xs   = work xs []

work     :: [a] → DList a
work xs = unwrap (body (wrap work)) xs
```

- Expand *unwrap*.

Reverse

Simplifying *rev*

```
rev      :: [a] → [a]
rev xs   = work xs []

work     :: [a] → DList a
work xs = fromList (body (wrap work) xs)
```

- Inline *body*.

Reverse

Simplifying *rev*

```
rev      :: [a] → [a]
rev xs   = work xs []

work     :: [a] → DList a
work xs = fromList (case xs of
    []      → []
    (x : xs) → wrap work xs ++ [x])
```

- case transformation.

Reverse

Simplifying *rev*

```
rev          :: [a] → [a]
rev xs       = work xs []
work         :: [a] → DList a
work []      = fromList []
work (x : xs) = fromList (wrap work xs ++ [x])
```

- *fromList* is morphism.

Reverse

Simplifying *rev*

$rev \quad \quad \quad :: [a] \rightarrow [a]$
 $rev\ xs \quad \quad = work\ xs\ []$
 $work \quad \quad \quad :: [a] \rightarrow DList\ a$
 $work\ [] \quad \quad = id$
 $work\ (x : xs) = fromList\ (wrap\ work\ xs) \circ fromList\ [x]$

- $fromList\ (wrap\ work\ xs)$
 $= unwrap\ (wrap\ work)\ xs$
 $= work\ xs$

Reverse

Simplifying *rev*

$$\begin{aligned} \text{rev} &:: [a] \rightarrow [a] \\ \text{rev } xs &= \text{work } xs [] \\ \text{work} &:: [a] \rightarrow \text{DList } a \\ \text{work } [] &= \text{id} \\ \text{work } (x : xs) &= \text{work } xs \circ \text{fromList } [x] \end{aligned}$$

- η -expand *work*.
- Expand *fromList* and \circ .

Reverse

Simplifying *rev*

```
rev           :: [a] → [a]
rev xs       = work xs []
work         :: [a] → [a] → [a]
work [] ys   = ys
work (x : xs) ys = work xs (x : ys)
```

- We have reached linear time accumulating version of list reversing function.

Memoisation

Our approach to memoising is to observe that any function f from natural numbers can be represented as infinite stream $[f\ 0, f\ 1, \dots]$.

unwrap/wrap

$unwrap \quad :: (Nat \rightarrow a) \rightarrow Stream\ a$

$unwrap\ f = map\ f\ [0..]$
 $\quad = f\ 0 : unwrap\ (f \circ (+1))$

$wrap \quad \quad :: Stream\ a \rightarrow (Nat \rightarrow a)$

$wrap\ xs = (xs!!)$

$(!!) \quad \quad :: Stream\ a \rightarrow Nat \rightarrow a$

$xs\ !!\ 0 \quad = head\ xs$

$xs\ !!\ (n + 1) = (tail\ xs)\ !!\ n$

Memoisation

We will show that $\text{wrap} \circ \text{unwrap} = \text{id}$ by expanding wrap and making arguments explicit: $(\text{unwrap } f) !! n = f \ n$.

- Base case:

$$\begin{aligned} & (\text{unwrap } f) !! 0 \\ &= (f \ 0 : \text{unwrap } (f \circ (+1))) !! 0 \\ &= f \ 0 \end{aligned}$$

- Inductive case:

$$\begin{aligned} & (\text{unwrap } f) !! (n + 1) \\ &= (f \ 0 : \text{unwrap } (f \circ (+1))) !! (n + 1) \\ &= \text{unwrap } (f \circ (+1)) !! n \\ &= (f \circ (+1)) \ n \\ &= f \ (n + 1) \end{aligned}$$

Fibonacci

Fibonacci function

fib :: *Nat* → *Nat*

fib = *fix body*

body :: (*Nat* → *Nat*) → *Nat* → *Nat*

body f 0 = 0

body f 1 = 1

body f (*n* + 2) = *f n* + *f* (*n* + 1)

Fibonacci

Simplifying *fib*

fib :: *Nat* → *Nat*

fib = *wrap work*

work :: *Stream Nat*

work = *unwrap* (*body* (*wrap work*))

- Apply *wrap* in *fib*.
- Apply *unwrap* in *work*.

Fibonacci

Simplifying *fib*

fib :: *Nat* → *Nat*

fib *n* = *work* !! *n*

work :: *Stream Nat*

work = *map* (*body* (*wrap* *work*)) [0..]

- Inline *body*.

Fibonacci

Simplifying *fib*

fib :: *Nat* → *Nat*

fib *n* = *work* !! *n*

work :: *Stream Nat*

work = *map* ($\lambda n \rightarrow$ **case** *n* **of**

0 → 0

1 → 1

(*n* + 2) → *wrap work n* + *wrap work* (*n* + 1)) [0..]

- Apply *wrap*.

Fibonacci

Simplifying *fib*

fib :: *Nat* → *Nat*

fib *n* = *work* !! *n*

work :: *Stream Nat*

work = *map* ($\lambda n \rightarrow$ **case** *n* **of**

0 → 0

1 → 1

(*n* + 2) → *work* !! *n* + *work* !! (*n* + 1)) [0..]

- Finally introduce *f*.

Fibonacci

Simplifying *fib*

fib :: *Nat* → *Nat*

fib *n* = *work* !! *n*

work :: *Stream Nat*

work = *map f* [0..]

where

f 0 = 0

f 1 = 1

f (*n* + 2) = *work* !! *n* + *work* !! (*n* + 1)

- We have reached a quadratic Fibonacci function from exponential one.

Continuations

- Any type can be alternatively represented as a continuation.
Idea is to represent value x as a function $\lambda c \rightarrow c\ x$ that takes a (continuation) c and applies it to x .

type $Cont\ a = (a \rightarrow a) \rightarrow a$

- We can convert from and to continuations as follows:

$toCont \quad :: a \rightarrow Cont\ a$

$toCont\ x \quad = \lambda c \rightarrow c\ x$

$fromCont \quad :: Cont\ a \rightarrow a$

$fromCont\ f = f\ id$

- It's easy to show that:

$$\begin{aligned}(fromCont \circ toCont)\ x &= (toCont\ x)\ id \\ &= (\lambda c \rightarrow c\ x)\ id \\ &= x\end{aligned}$$

Evaluation

- We will now consider simple expression language:

```
data Expr = Val Int
          | Expr ⊕ Expr
          | Throw
          | Catch Expr Expr
```

- With standard evaluation function:

```
eval      :: Expr → MInt
eval (Val n)      = Just n
eval (e0 ⊕ e1)   = case eval e0 of
                    Nothing → Nothing
                    Just n   → case eval e1 of
                                Nothing → Nothing
                                Just m   → Just (n + m)
eval Throw      = Nothing
eval (Catch e0 e1) = case eval e0 of
                    Nothing → eval e1
                    Just n   → Just n
```

Evaluation

- One might expect to move to representation of type $Expr \rightarrow Cont\ MInt$. Instead we will have different continuation for exceptional control flow and regular control flow.
- We split $MInt \rightarrow MInt$ into two and reach type:

$$Expr \rightarrow (Int \rightarrow MInt) \rightarrow MInt \rightarrow MInt$$

- Wrap/unwrap are given by:

$$\begin{aligned} unwrap\ g\ e\ s\ f &= \text{case } g\ e\ \text{of} \\ &\quad Nothing \rightarrow f \\ &\quad Just\ n \rightarrow s\ n \\ wrap\ h\ e &= h\ e\ Just\ Nothing \end{aligned}$$

Evaluation

- Worker/wrapper assumption can easily be verified as follows:

$$\begin{aligned} & (\text{wrap} \circ \text{unwrap})\ g\ e \\ &= \text{wrap}\ (\text{unwrap}\ g)\ e \\ &= \text{unwrap}\ g\ e\ \text{Just Nothing} \\ &= \text{case } g\ e\ \text{of} \\ &\quad \text{Nothing} \rightarrow \text{Nothing} \\ &\quad \text{Just } n \rightarrow \text{Just } n \\ &= g\ e \end{aligned}$$

- We can now apply the worker/wrapper transformation.

Evaluation

Simplifying *eval*

$eval :: Expr \rightarrow MInt$

$eval = wrap\ work$

$work :: Expr \rightarrow (Int \rightarrow MInt) \rightarrow MInt \rightarrow MInt$

$work = unwrap\ (body\ (wrap\ work))$

- Inline *wrap*.
- η -expand *work*.

Evaluation

Simplifying *eval*

eval $:: Expr \rightarrow MInt$

eval $= work\ e\ Just\ Nothing$

work $:: Expr \rightarrow (Int \rightarrow MInt) \rightarrow MInt \rightarrow MInt$

work $e\ s\ f = unwrap\ (body\ (wrap\ work))\ e\ s\ f$

- Apply *unwrap*.

Evaluation

Simplifying *eval*

$eval \quad :: Expr \rightarrow MInt$
 $eval \quad = work\ e\ Just\ Nothing$
 $work \quad :: Expr \rightarrow (Int \rightarrow MInt) \rightarrow MInt \rightarrow MInt$
 $work\ e\ s\ f = \text{case } body\ (wrap\ work)\ e\ \text{of}$
 $Nothing \rightarrow f$
 $Just\ n \rightarrow s\ n$

- Apply *body*.

Evaluation

Simplifying *eval*

```
work e s f = case (case e of
    Val n      → Just n
    e0 ⊕ e1   → case wrap work e0 of
    Nothing → Nothing
    Just n   → case wrap work e1 of
    Nothing → Nothing
    Just m  → Just (n + m)
    Throw   → Nothing
    Catch e0 e1 → case wrap work e0 of
    Nothing → wrap work e1
    Just n  → Just n) of
    Nothing → f
    Just n  → s n
```

Evaluation

Simplifying *eval*

work e s f = **case** e **of**

Val n $\rightarrow s$ n

$e_0 \oplus e_1 \rightarrow$ **case** *wrap* *work* e_0 **of**

Nothing $\rightarrow f$

Just $n \rightarrow$ **case** *wrap* *work* e_1 **of**

Nothing $\rightarrow f$

Just $m \rightarrow s$ ($n + m$)

Throw $\rightarrow f$

Catch e_0 $e_1 \rightarrow$ **case** *wrap* *work* e_0 **of**

Nothing \rightarrow **case** *wrap* *work* e_1 **of**

Nothing $\rightarrow f$

Just $n \rightarrow s$ n

Just $n \rightarrow s$ n

Evaluation

Simplifying *eval*

$work (Val\ n)\ s\ f \quad =\ s\ n$
 $work (e_0 \oplus e_1)\ s\ f \quad =\ \mathbf{case\ wrap\ work\ } e_0\ \mathbf{of}$
 $Nothing \rightarrow f$
 $Just\ n \rightarrow \mathbf{case\ wrap\ work\ } e_1\ \mathbf{of}$
 $Nothing \rightarrow f$
 $Just\ m \rightarrow s\ (n + m)$
 $work\ Throw\ s\ f \quad =\ f$
 $work (Catch\ e_0\ e_1)\ s\ f =\ \mathbf{case\ wrap\ work\ } e_0\ \mathbf{of}$
 $Nothing \rightarrow \mathbf{case\ wrap\ work\ } e_1\ \mathbf{of}$
 $Nothing \rightarrow f$
 $Just\ n \rightarrow s\ n$
 $Just\ n \rightarrow s\ n$

Evaluation

- By re-exposing *unwrap* we can perform further simplification:

$$\begin{aligned} & \mathbf{case} \text{ wrap work } x \mathbf{ of} \\ & \quad \text{Nothing} \rightarrow g \\ & \quad \text{Just } n \rightarrow s \ n \\ = & \quad \{ \text{unapply } \text{unwrap} \} \\ & \quad \text{unwrap} (\text{wrap work}) \times s \ g \\ = & \quad \{ \text{worker/wrapper fusion} \} \\ & \quad \text{work } \times s \ g \end{aligned}$$

- We can apply this multiple times.

Evaluation

Simplifying *eval*

$work\ (Val\ n)\ s\ f \quad =\ s\ n$

$work\ (e_0 \oplus e_1)\ s\ f \quad =\ work\ e_0\ (\lambda n \rightarrow$
 $\quad\quad\quad work\ e_1\ (\lambda m \rightarrow s\ (n + m))\ f)\ f$

$work\ Throw\ s\ f \quad =\ f$

$work\ (Catch\ e_0\ e_1)\ s\ f \quad =\ work\ e_0\ s\ (work\ e_1\ s\ f)$

- Corresponds to abstract machine that works on two stacks, one for normal evaluation and other for handling exceptions.

Fin.